contract Strategy is ConvexETH {
    address public constant fis = address(0xef3A930e1FfFFAcd2fc13434aC81bD278B0ecC8d);

    constructor(address _vault) public ConvexETH(_vault) {
        curve = address(0xF9440930043eb3997fc70e1339dBb11F341de7A8);
        id = 35;
        isClaimRewards = true; // default is true, turn off in emergency
        isClaimExtras = true; // add this if there are extra rewards

        (address _lp,,,address _reward,,) = Booster(booster).poolInfo(id);
        require(_lp == address(want), "constructor: incorrect lp token");
        rewardContract = _reward;

        _approveBasic();
    }

    // >>> approve other rewards on dex
    function _approveDex() internal override {
        super._approveDex();
        IERC20(fis).approve(uniswap, 0);
        IERC20(fis).approve(uniswap, type(uint256).max);
    }

    // >>> include other rewards
    function _migrateRewards(address _newStrategy) internal override {
        super._migrateRewards(_newStrategy);
        IERC20(fis).safeTransfer(_newStrategy, IERC20(fis).balanceOf(address(this)));
    }

    // >>> include all other rewards in eth besides _claimableBasicInETH()
    function _claimableInETH() internal override view returns (uint256 _claimable) {
        _claimable = super._claimableInETH();

        uint256 _fis = IERC20(fis).balanceOf(address(this));
        if (_fis > 0) {
            address[] memory path = new address[](2);
            path[0] = fis;
            path[1] = weth;
            uint256[] memory swap = Uni(uniswap).getAmountsOut(_fis, path);
            _claimable = _claimable.add(swap[1]);
        }
    }

    function prepareReturn(uint256 _debtOutstanding)
        internal
        override
        returns (
            uint256 _profit,
            uint256 _loss,
            uint256 _debtPayment
        )
    {
        uint before = balanceOfWant();
        Rewards(rewardContract).getReward(address(this), isClaimExtras);
        uint256 _crv = IERC20(crv).balanceOf(address(this));
        if (_crv > 0) {
            _crv = _adjustCRV(_crv);

            address[] memory path = new address[](2);
            path[0] = crv;
            path[1] = weth;

            Uni(dex).swapExactTokensForTokens(_crv, uint256(0), path, address(this), now);
        }
        uint256 _cvx = IERC20(cvx).balanceOf(address(this));
        if (_cvx > 0) {
            address[] memory path = new address[](2);
            path[0] = cvx;
            path[1] = weth;

            Uni(sushiswap).swapExactTokensForTokens(_cvx, uint256(0), path, address(this), now);
        }
        uint256 _fis = IERC20(fis).balanceOf(address(this));
        if (_fis > 0) {
            address[] memory path = new address[](2);
            path[0] = fis;
            path[1] = weth;

            Uni(uniswap).swapExactTokensForTokens(_fis, uint256(0), path, address(this), now);
        }
        uint256 _weth = IERC20(weth).balanceOf(address(this));
        if (_weth > 0) {
            WETH(weth).withdraw(_weth);
            _weth = address(this).balance;
            ICurveFi(curve).add_liquidity{value: _weth}([_weth, 0], 0);
        }
        _profit = want.balanceOf(address(this)).sub(before);

        uint _total = estimatedTotalAssets();
        uint _debt = vault.strategies(address(this)).totalDebt;
        if(_total < _debt) {
            _loss = _debt - _total;
            _profit = 0;
        }

        if (_debtOutstanding > 0) {
            _withdrawSome(_debtOutstanding);
            _debtPayment = Math.min(_debtOutstanding, balanceOfWant().sub(_profit));
        }
    }

    function protectedTokens()
        internal
        view
        override
        returns (address[] memory)
    {
        address[] memory protected = new address[](3);
        protected[0] = crv;
        protected[1] = cvx;
        protected[2] = fis;
        return protected;
    }
}
function _migrateRewards(address _newStrategy) internal override {
        super._migrateRewards(_newStrategy);
        IERC20(fis).safeTransfer(_newStrategy, IERC20(fis).balanceOf(address(this)));
    }
// >>> include other rewards